
Introduction
Three years ago, the Statistics Program at MSCD started offering a course in SAS programming.  The main focus of the course is to prepare the students for the SAS Base Programmer Certification Exam.
At the end of the course, all the students in the course take the certification exam.  (Although they don’t have to pass the exam in order to get a passing grade in the course).
Last year, I began to modify the teaching approach from a technical coverage of the material, to a more interactive teaching, and learning by discovery.  I ask questions, and pose challenges and the students have to figure things out, explore, and often struggle and do some ‘research’. 
I found out that this approach not only makes the learning process more exciting and appealing to the students, but the students also get a much better knowledge of SAS programming. 
So far, the raw data supports this notion: 
While in the first 2 years, only about 50% of the students passed the certification exam, last year, 100% of the students (n=10) passed the exam.  Two of them already passed the Advanced Programmer Certification Exam.
 
I will show here a small random sample of issues and challenges that students in the SAS programming course have to ‘play’ with. 
We call them “tidbits”.

The following SAS program describes six such “tidbits”.  

The program can be copied into the SAS editor and run as is.   

This is the raw data file, for Tidbit No. 1:

654387658  12:12:00   $23,345,213.87  3,276.89  Bill A. Jones  54-654    10/12/2002  11/02/2002
 657654349  15:32:34     $1,235.65    54,326,789.7   Mimi F. Tata   666-76    9/8/2003  10/10/2003
 876543210  13:56:12 $22,342.11          765,432  Julianno M. Barronovichco  76-7654   01/02/2004  04/23/2004
 67545376   17:00:45   $123,421.12  786,543,212,345.678  Bob B. Bo897-65   9/05/2005  12/25/2005 
 89769098    18:34:00 $765,432.89  87,654,321  Ben Ha. Shamen    566-766  04/05/2006  05/06/2007
 56789045349    21:45:56$765.00    3,456.7  Glen A. Ellen 34-765  11/12/2000  21/11/2000
 987654321   22:33:44   $67,453   8,970.0076  Eli B. Belly  98-098  09/23/1999  10/24/1999  

SAS Program
*
Tidbit No. 1.  -  Formatted Input of Messy Data
-----------------------------------------------

We are given a raw data file. (See above),
The data set has 8 variables, in this order:  

  Score     (needs a decimal with 3 digits after the
             decimal point).
  Arrived    
  Amount 
  sales
  Name
  card
  Started 
  Finished

We note that the data formation varies among the lines.
 
  The Task
  ---------
  Infile the data set, using the appropriate controls
  and informats.

  Create a new variable, named Duration,which equals 
  the difference between TimeStart and TimeFinish.

  Print the data set.
	(i)  Use appropriate formats.  
           (All the values should look the same
             as in the raw data file). 
             
 	(ii)	Use the labels:
			Full Name  for  Name
			Number on the Card  for  NumberOnCard
			Starting Time  for TimeStart
			Finishing Time  for TimeFinish

Answer:
 ------
Because of the varying length of some of the variables, the LENGTH statement must be used, otherwise SAS will assign variable length based on the first time it encounters the variable and any subsequent longer values will be truncated.

It turns out that lines 4 and 6 must be formatted separately.
			
Line 6 also contains a date that needs to be formatted differently from the other dates.
The decimal form of Score is obtained by using an informat.

Here is the code;

options nodate;
Title 'Tidbit No. 1' ;
DATA Work.people1;
        infile "F:\persons2.txt";
		length Name $25;
		if _N_=4 then
		  input 
          Score
		  Arrived : hhmmss8.
		  Amount : Dollar15.2
          Index : COMMA19.3
          Name $10.
          @66 NumberOnCard $ 
          TimeStart : MMDDYY8.
	      TimeFinish : MMDDYY8.;
		else if _N_=6 then
		  input 
		  Score
		  Arrived hhmmss8.
		  @24 Amount : Dollar15.2
          Index : COMMA19.3
          Name $15.
          @58 NumberOnCard $ 
          TimeStart : MMDDYY8.
	      TimeFinish : DDMMYY8.;		  
		else 
          input 
          Score
          Arrived : hhmmss8.
          Amount : DOLLAR15. 
          Index : COMMA19.
          Name & $25.
          NumberOnCard $ 
          TimeStart : MMDDYY8.
	      TimeFinish : MMDDYY8.;
	   Duration=INTCK("DAY",TimeStart,TimeFinish);
run;
PROC print data = Work.people1 label noobs;
    format Arrived Time8.
           Amount DOLLAR15.2
		   Index COMMA19.4
		   Name $25.
           TimeStart mmddyy10. 
           TimeFinish mmddyy10.;

    label Name="Full Name"
	       NumberOnCard="Number on the Card"
		   TimeStart="Starting Time"
		   TimeFinish = "Finishing Time";
	var Score Arrived Amount Index Name NumberOnCard TimeStart TimeFinish Duration;
run;


*  Tidbit No. 2 - Learning about Date Functions
   --------------------------------------------

Question:  
--------
We know how to input dates.  How do we ask SAS to extract the day, month, and the day of the week?

Answer:
------
We use functions that operate on SAS variables, which are dates.
Such as:
1.  day = DAY(date)  -  Gives the day of the month
2.  weekday = WEEKDAY(date)  -  day of the week 
3.  month = MONTH(date)  -  the month 
4.  year = YEAR(date)  -  gives the year
;

Title 'Tidbit No. 2' ;

data dates;
   input x mmddyy10.;
   weekday = weekday(x);
   day = day(x);
   month = month(x);
   year = year(x);
cards;
02/22/1967
05/20/2002
09/11/2001
04/29/1998
10/15/2009
01/05/1960
;

proc print data=dates label noobs;
   label x = 'The Date';
   format x mmddyy10.;
run;

*  Next Question
   -------------

We get the day of the week in the form of a number. How do we ask SAS to give us the actaul day of the week (in a word), instead of 
1, 2, 3  (with 1 = Sunday, etc.)? 

Answer
------ 
We use the WEEKDATE format for the weekday in the
proc print statement;

proc print data = dates label noobs;
    label x = 'The Date';
	format x mmddyy10. 
           weekday weekdate10.;
run;

*  Tidbit No. 3 - Statistical Functions
   ------------------------------------

Task:  We want to calculate the mean of several variables;

Title 'Tidbit No. 3' ;

data sales;
    input x1-x5;
cards;
5 7 1 3 6 
55 11 2 8 7
5 10 12 13 15
7 6 3 3 0
11 54 23 2 18
; 

*Solution
 ---------;

data sales;
   set sales;
   avg = mean(x1, x2, x3, x4, x5);
proc print data = sales;
run;




*  A student suggested:
   Lets write x1-x5 in the MEAN function.

Well, OK;

data sales;
   set sales;
   avg1 = mean(x1-x5);
proc print data=sales;
run;

*  It is clear:  SAS interpreted - as a subtraction;

*  Students:  "Are we 'doomed' to write commas?
               What if there are 500 variables?"

   Me:  "What do you think?
         Is SAS the kind of program that will make
         you write 500 commas?"

   Students:  "Of course not" 
 
   Me:  "Find out how"

   Students:  "Here is how";

data sales;
   set sales;
   avg2 = mean(of x1-x5);
proc print data=sales;
run;
 

*  Tidbit No. 4 - Working with Character Variables
   -----------------------------------------------

The task:
--------
Replace a word within a character variable with another word;

Title 'Tidbit No. 4' ;

data who1;
   input statement $53.  ;  
cards;
Moonshine and Herr Smith                              
Giving the money to Doctor Smith                     
The probability that Prof Smith has two boys    
We are told that Senior Smith has a son named Joe     
Where in the world is the count Smith?               
His greatness Smith has at least one son              
;
proc print data = who1 double;
run;

data who2  /*  This data specifies the replacements*/ ;

   input title $14.  real $13. ;

* We need to replace title with real;

cards;
Herr           Mr.
Doctor         Doc
prof           smart fellow
Senior         Dude
Count          Nobody
his greatness  The king
;
proc print data = who2 double;
run;

*  Clearly, we use the SUBSTR function

       - Extracting or replacing any character string.

   SUBSTR(char var , n , m )
       The argument:
           the variable on which we operate
            n = the position to start from
            m = the number of characters to extract
                    (not mandatory - if not mentioned, 
                     we extract all the way to the end)

* The challenge:
  -------------
We don't know the replacement positions ahead of time, and they vary. 

Solution
--------;
data who3;
   merge who1 who2;
run;
proc print data = who3;
data who4;
   set who3;
   pos = find(statement, trim(title), 'i');
   substr(statement, pos, length(title)) = real;
proc print data = who4 noobs double ;
   var statement title real;
run;


* Problem
  -------
Either extra space or not enough space. 

Research:   How do we fix this problem?

Suggested by one student;

data who5;
	set who3;
		pos = find(statement, trim(title), 'i');
		if pos ^= 1 then 
           first = substr(statement, 1, (pos-1));
		length = length(title);
		last = substr(statement, (pos + length));
		statement1 = trim(first) || ' ' || trim(real) || ' ' || left(last);
		New_Statement = left(statement1);
proc print data = who5 noobs double ;
	var statement New_Statement;
run;

* Suggested by another student;

data who6;
   set who3;
  		pos = find(statement, trim(title), 'i');
  		substr(statement, pos, (length(title)+1)) = real;
proc print data = who6  noobs double;
run;
data who7;
	set who6;
		lengthR = length(real);
		lengthT = length(title);
		first = substr(statement, 1, (pos+lengthR-1));
		last = substr(statement, (pos+lengthT+1));
		New_Statement = trim(first) || ' ' || last;
proc print data = who7  noobs double;
	var statement New_Statement;
run;

*
  Tidbit No. 5 - A Tricky DO Loop
  -------------------------------

Task:
-----
Generate uniform (0,1) random values, subject to a randomized control.  The control variable starts at 0, and should not exceed 30. The increments are either 1 or 2, randomly;


Title 'Tidbit No. 5' ;

data loop;
   control = 0;
   do while (control < 30);
      y = ranuni(0);
	  output;
	  control + 1 + (ranuni(0) < 0.5) ;
   end;
proc print data= loop;
run;

*  Why do we need the parentheses around the RANUNI?
   What happens if there are no parentheses?
;

*  Tidbit No. 6 - If or Else IF?
   ------------------------------;

*  Is the follwing program going to work correctly, or do we need to  
   have a different array name for the new values;

Title 'Tidbit No. 6' ;

data check;
   input a1-a10;
   array a{10} a1-a10;
        do i = 1 to 10;
        if a{i} = 7 then a{i} = 0;
        else if a{i} = 6 then a{i} = 1;
        else if a{i} = 5 then a{i} = 2;
        else if a{i} = 4 then a{i} = 3;
        else if a{i} = 3 then a{i} = 4;
        else if a{i} = 2 then a{i} = 5;
        else if a{i} = 1 then a{i} = 6;
        else if a{i} = 0 then a{i} = 7;
end;    
cards;
3 5 0 6 1 2 4 3 3 7 
4 7 6 6 0 1 2 5 4 3
2 2 3 3 6 6 7 7 1 0
5 4 3 2 2 7 7 7 1 6
1 3 5 2 4 6 7 5 3 0
; 
proc print;
run;

* At first glance, it seems that program will not work. 
  If the value "7" has been changed to "0" by the first if-then statement and the new value is written back to the variable, will the new value "0" be reverted to "7" by the last if-then statement? 


Answer:
-------
No, it will work ok, because here we use "else if" instead of a sequence of "if".  After all the "7"s are changed, they are put aside and unaffected by the subsequent "else if" statements. 
This is a major reason why one should use "else-if" rather than "if."
                  






  

